# UNIVERSITY OF CALIFORNIA OBSERVATORIES
# LICK OBSERVATORY

---

TECHNICAL REPORT

## DEIMOS TV Guider Optical Design

Brian M. Sutin

April 3, 1997

University of California Observatories/Lick Observatory
University of California Santa Cruz
Santa Cruz, California, 95064

#83

This technical report describes the current design for the DEIMOS TV guider. The optical train consists of a Photometrics camera with a 24-micron, 2K x 2K CCD detector with a Canon 200mm f/1.8 lens and a single folding flat mirror, looking at reflective surfaces near the focal surface of the telescope. The final scale is 0.2 arcsec per pixel, for a total TV guider view of 11.63 square arcmin. 4.5 square arcmin of the field of view is always directly viewing the sky for offset guiding. This area at the telescope focal surface is occupied by a 2.3 inch by 5.9 inch mirror with a concave curvature of 85.0 inches. The remaining area of the field of view either images the cylindrical slit mask while DEIMOS is in multi-slit spectrographic mode, or views a spherical plate with a long slit. This spherical plate has the same curvature as the offset guiding mirror. The curves on the reflective elements at the focal surface serve to put the pupil halfway between the Canon lens aperture and the flat fold mirror.

In order to fit the TV system into the instrument package, a 4.0 inch by 5.7 inch flat folding mirror is placed approximately 41.5 inches above the telescope focal surface, and 1.2 inches off the central instrument axis. This mirror sits at the optimal location to catch light from the multi- slit masks, while not blocking light traveling from the telescope towards the telescope focal surface.

The figures and appendices included in this report are as follows. The first two pages are the original design notes for the TV system, which are probably too cryptic for anyone except the author to fathom. Because of the effect of telescope field distortion, the actual final design differs from the design notes by various fudge factors which run on the order of a degree or less in angle, and a few tenths of an inch in space. Throughout the design, the full Keck aperture diameter was assumed. The next nine pages are the command descriptions for the Keck telescope, the TV system, and the Canon lens. For the design, the Canon was focussed at approximately the midpoint of the focus travel.

The next four figures are crude pictures of the TV layout. The first is a view down the instrument axis, showing the various parts of the optical train in offset-guiding mode, with a few rays drawn in. The following three pictures are side views of the TV layout in the three modes.

The next four figures show the vignetting and RMS spot sizes for the various TV modes. The first figure shows the vignetting for a cylindrical multi-slit mask. The vignetting is negligible near the nominal center of the field, at 4.5 arcmin from the telescope axis. The vignetting in the offset guiding mode and the long-slit mode are less than one percent, not including any internal vignetting from either the Canon lens (when off-axis) or the Photometrics camera. The next three figures show the RMS spotsize diameter in all three modes. These figures only include the effect of defocus, due to the curved field of view of the telescope and the off-axis position of the TV system. Most of the defocus effects have been compensated by tilting the reflecting elements at the telescope focal surface, and by tilting the detector chip. Aberrations due to the Canon lens are unknown.

The next two figures show the placement of the light beams at the Canon lens aperture. Since the circles represent the full Keck aperture beam, the amount of light missing the aperture is clearly small. A similar amount of light is lost at the flat fold mirror, which is approximately the same size.

WANT $0.2''/\text{PIX}$ HAVE $24\mu$ pixels

FIELD SCALE $\approx 728\mu/''$ $\Rightarrow$ WANT DEMAG OF 6.07

$\Rightarrow$ PARAXIAL DISTANCE OF $6.07 = \frac{T-F}{F} \Rightarrow T = 55.64''$

APERTURE DISTANCE IS $55.64'' - \approx 180\text{mm} = 48.56''$

↑ PRINCIPLE PLANE DISTANCE

TOTAL FOV IS $3.41' \times 3.41'$.




$4.5' = 7.74''$ IN FOCAL SURFACE

$\Rightarrow \theta = 0.562°$ (INCOMING LIGHT ANGLE)




$\phi = 12° - \theta = 11.44°$ - OUTGOING LIGHT ANGLE

$\tan\phi = \frac{7.74'' + D}{48.56''} \Rightarrow D = 2.09''$ ~ DISTANCE OFF CENTRAL AXIS.




X: $\frac{3.41'}{2} \Rightarrow 2.93'' \Rightarrow \theta_x = \tan^{-1}\frac{2.93''}{48.56''} = 3.45°$

Y: $5.0' - \frac{3.41'}{2} \Rightarrow 3.295'' \Rightarrow \theta_y = \tan^{-1}\frac{3.295'' + 2.09''}{48.56''} = 6.33°$

MIRROR TILT $= \frac{1}{2}\tan\frac{2.09}{48.56} = 1.23°$ ✓

$48.56'' - 7'' = 41.56''$ TO MIRROR

BEAMSIZE @ MIRROR IS 3.63" DIAMETER

DSIZE IS 0.84"

$\Rightarrow$ MIRROR DIAMETER $\approx$ 3.88" DIAMETER

USE 4" NEWTONIAN ELLIPTICAL FLAT




LIC CENTRE IS AT 0.77" $\Rightarrow$ 1.23" FROM CENTER

INCOMING BEAM AT $2.73'' - 1.52'' = 1.21''$ $\Rightarrow$ NO VIGNETTING

BEAM AT 3.1" $\Rightarrow$ OK - NO VIGNETTING.

BEAM @ CAMERA IS $\frac{48.56}{13.7} = 3.54''$

HAVE $2'' - \frac{3.54''}{2} = 0.23''$ FREE PLAY, AT CAMERA MOUTH

.23" ⇒ ≈ 0.13" ON EACH SIDE







$\left(\frac{34}{788.3}\right) = .0432 = \tan\theta$

$\times 41.56 = 1.80$

$5.33 - (1.8) = 3.533 - 0.77 = 2.76''$

```
#
#  KECK_TELESCOPE -- describes the Keck Telescope
#
#  Brian Sutin -- Last Change July 24, 1996
#
#  The following is a description of the current model of the Keck I
#  and Keck II telescopes.  It should correspond to the as-build values.
#  Most of these numbers come from a xerox of KOTN #163, with corrections
#  in the margins, or were gotten from Jack Osborne.
#
#  All units are in (yuk) inches.
#
#  Pullback is the number of inches forward (away from the primary) that
#  the focal surface is moved from the nominal back distance.  The
#  secondary is pistoned the appropriate amount to correct for the change.
#

Aperture.Radius = 215.5309557          # outer tips of hexes
Aperture.Hole   =  52.165              # shroud radius
Aperture.Keck   = TRUE                 # Apodize over the irregular edge

KeckPerfect     = FALSE                # Use paraxial approximation
KeckTertiary    = FALSE                # Add tertiary for pictures

KeckFieldRadius = 85.5                 # radius of field curvature.

WhichKeck       = KeckII

KeckI  = 1
KeckII = 2

DEFINE KECK_TELESCOPE( PULLBACK )
    {
    KeckISC   =  186.53685             # from Epps
    KeckIISC  =  186.45021             # from Epps 7/24/96
    KeckISA2  = -1.6436145             # from Epps
    KeckIISA2 = -1.6413676             # from Epps 7/24/96

    if( WhichKeck == KeckI )
        {
        Src = KeckISC
        SA2 = KeckISA2
        }
    elseif( WhichKeck == KeckII )
        {
        Src = KeckIISC
        SA2 = KeckIISA2
        }

#
#  The following are the quantities needed for computing the focus.
#  Dps is the Primary-Secondary distance.
#

    Prc   = 1376.92913                 # Primary Radius of Curvature
    Dback =    98.425197 + PULLBACK    # Back Distance
    B = Prc - 2 * ( Src + Dback )                      # temporary
    C = Prc * Src + 2 * Dback * ( Prc - Src )          # temporary
    Dps = ( B + sqrt( B * B + 4 * C ) ) / 4

#
#  The Secondary Shroud
#
#  The secondary shroud blocks light from hitting the secondary from
```

```
# directions other than the primary.  This surface describes the
# shadow which the shroud blocks from the beam coming into the primary.
# If the shroud is not deployed, the baffle is hexagonal with a
# minimum radius of 45.176 inches.  From a conversation with Jerry
# Nelson.
#

    ZMOV( -565.9432 )
        {
        LABEL( "Keck Secondary Shroud" )
        FLAT
        STOPCIRCLE( -52.165 )
        }

#
# The Primary Mirror
#
# The primary mirror is represented as a continuous surface.  Ha!
# The central hole is obscured by the shroud, so is not represented.
#

    LABEL( "Keck Primary Mirror" )
    STOPCIRCLE( 215.5309557 )                   # outer tips of hexes
    IF( KeckPerfect )
        {
        PARAX( Prc / 2 )
        FLAT
        ZMOV(0)
        FLAT
        }
    ELSE
        {
        CONIC( -1 / Prc, -1.003683 )
        }
    MIRROR                                      # primary

#
# The Secondary Mirror
#
# These are the as-built Keck I secondary values.  I have no idea
# where the radius value came from, but it is larger than the value
# required for a 10 arcminute field of view.
#

    {*
    ZMOV( -Dps )                                # primary-secondary distance
    XROT( 180 )
    *}
        {
        LABEL( "Keck Secondary Mirror" )
        STOPCIRCLE( 28.15 )
        IF( KeckPerfect )
            {
            PARAX( -Src / 2 )
            FLAT
            ZMOV(0)
            FLAT
            }
        ELSE
            {
            CONIC( 1 / Src, SA2 )
            }
        MIRROR
        }
```

```
#
#   The Tertiary Mirror
#
#   The tertiary would go here if we bothered.  It is 4.0 meters in
#   front of the primary.
#
#   I got the size (1.068 x 1.439) from Jerry.   This is too small.
#   I hope that the real tertiary is larger.
#

    if( KeckTertiary )
        {
        ZMOV( -4.0 / 0.0254 )
            {
            XROT( 45 )
            LABEL( "Keck Tertiary Mirror" )
            MIRROR
            FLAT
#           STOPRECT( 0.5 * 1.068 / 0.0254, 0.5 * 1.439 / 0.0254 )
            STOPRECT( 0.5 * 1.062 / 0.0254, 0.5 * 1.079 * sqrt(2) / 0.0254 )
            XROT( 180 + 45 )
            }
        }

#
#   The Focal Surface
#
#   This is the focal surface of the telescope, where the slit mask
#   for the spectrograph goes.  The radius of curvature here gives
#   round spots.  Something like 83.6 gives a slightly smaller
#   elliptical spot.
#
#   The radius of the aperture is from assuming that there is a mask
#   to only allow light from a field of 10 arcminutes.
#   10 arcminutes corresponds to 17.095 inches. [???]
#

    ZMOV( Dback )                       # back distance
    LABEL( "Keck Focal Surface" )
    if( KeckPerfect )
        {
        FLAT
        }
    ELSE
        {
        SPHERE( 1.0 / -KeckFieldRadius )
        }
#   STOPCIRCLE( 17.11 )                 # there is no real stop here!
    }
```

```
#!/u/sutin/bin4/sp4

###########################################################################
#
#  TV -- the DEIMOS TV System
#
#  Brian Sutin, 03/19/96
#
#  All units are inches.
#
#  The various fudge factors are due to the telescope distortion.
#  I used 1.72 in/arcmin to place objects at the focal surface, but
#  this value is quite wrong near the center.  I should have made a
#  distortion polynomial fit, and then inverted it for object placement.
#

###########################################################################
#
#  External Include Files
#

FILE "BASIC"                            # basic internal stuff
FILE "KECK"                             # keck telescope
FILE "CANON"

###########################################################################
#
#  Operations
#

MAKE_PHYS       = 0                     # 2/general 4/imaging 5/tv
MAKE_BEAM       = 3                     # 3/beam cross sections
MAKE_SPOTS      = 0                     # 5/general
MAKE_AUTOCAD    = 2                     # 1 or 2

PlotAxis        = 0
OpticNoStops    = 0
DELTA           = 4

AUTOCADUNIT     = 1

Mode            = ModeMultiSlit
# Mode          = ModeLongSlit
# Mode          = ModeDirectView

EDGEKLUDGE      = 0.1                   # extra edge slop

if( MAKE_AUTOCAD == 1 )
    {
    AutoCAD.File = "autocad/" // autoname // "optc.scr"
    }
elseif( MAKE_AUTOCAD == 2 )
    {
    AutoCAD.File = "autocad/" // autoname // "rayc.scr"
    }

#
#  TV Field
#
#  The TV images a region 3.41 x 3.41 arcmin square.  One edge is
#  against the symmetry axis, while the other is 0.5 arcmin above
#  the nominal slit position of 4.5 arcmin.
#

if( Mode == ModeMultiSlit )
```

```
    {
    FieldHalfWidthX      = 3.41 / 2
#   FieldHalfWidthY      = 0.5                    # serious viewing at 4.5'
    FieldHalfWidthY      = (5.0 - 3.1) / 2        # large area viewing

    FieldOffsetX         = FieldHalfWidthX
    FieldOffsetY         = 5.0 - FieldHalfWidthY

    autoname             = "slit"
    }
elseif( Mode == ModeLongSlit )
    {
    FieldHalfWidthX      = 3.41 / 2
    FieldHalfWidthY      = ( 5.0 - 3.1 ) / 2

    FieldOffsetX         = FieldHalfWidthX
    FieldOffsetY         = 5.0 - FieldHalfWidthY

    autoname             = "long"
    }
elseif( Mode == ModeDirectView )
    {
    FieldHalfWidthX      = 3.41 / 2
    FieldHalfWidthY      = ( 2.9 - (5.0 - 3.41) ) / 2

    FieldOffsetX         = 3.41 / 2
    FieldOffsetY         = 2.9 - FieldHalfWidthY

    autoname             = "view"
    }

SpotCount       = 500
Wavelength      = 0.5

ModeMultiSlit   = 1
ModeDirectView  = 2
ModeLongSlit    = 3

######################################################################
#
#  Internal Global Variables
#

WLMIN                    =   0.3900      # minimum wavelength
WLMAX                    =   1.1000      # maximum wavelength

######################################################################
#
#  DEIMOS TV Optical Description
#

#
#  First the telescope, with the 3.0-inch pullback
#

KECK_TELESCOPE( 3 )
STOPCIRCLE( 17.2 )                       # for drawing purposes

if( MAKE_SPOTS )
    {
    VIEW
    }

KeckFieldScale = 1.72                    # inches/arcmin
```

```
########################################################################
#
#  The slit mask surface
#

SLIT_CURV = 81.57
LONG_SLIT_CURV = 85.0
FieldRC = 85.5

if( Mode == ModeMultiSlit )
    {
    {*
    ZMOV( 0.4 )
    XROT( -6.0 )
    YMOV( 5.32 + ( 8.562 / 2 ) )                                  # D1705.K
    *}
        {
        LABEL( "DEIMOS Slit Mask Surface" )
        CYLINDER( -1.0 / SLIT_CURV )
        STOPRECT( 29.802 / 2, 8.562 / 2 )                         # D1705.K
        MIRROR
        }
    }
elseif( Mode == ModeLongSlit )
    {
    {*
    ZMOV( -FieldRC )
        {
        YROT(  atan( FieldOffsetX * KeckFieldScale / FieldRC ) + 0.23 )
        XROT( -atan( FieldOffsetY * KeckFieldScale / FieldRC ) - 0.01 )
        }
    XROT( -0.90 )                        # fudge factor
    ZMOV( -0.062 )                       # fudge factor
    *}
        {
        LABEL( "DEIMOS Long Slit Surface" )
#       SPHERE( -1 / SLIT_CURV )
        SPHERE( -1 / LONG_SLIT_CURV )
        STOPRECT( FieldHalfWidthX * KeckFieldScale + EDGEKLUDGE,
                  FieldHalfWidthY * KeckFieldScale + EDGEKLUDGE )
        MIRROR
        }
    }
elseif( Mode == ModeDirectView )
    {
    {*
    ZMOV( -FieldRC )
        {
        YROT(  atan( FieldOffsetX * KeckFieldScale / FieldRC ) )
        XROT( -atan( FieldOffsetY * KeckFieldScale / FieldRC ) )
        }
    XROT( -1.0 )                         # fudge factor
    ZMOV( -0.065 )                       # fudge factor
    *}
        {
        LABEL( "DEIMOS TV Direct Viewing Mirror" )
#       SPHERE( -1 / SLIT_CURV )
        SPHERE( -1 / LONG_SLIT_CURV )
        STOPRECT( FieldHalfWidthX * KeckFieldScale + EDGEKLUDGE,
                  FieldHalfWidthY * KeckFieldScale + EDGEKLUDGE )
        MIRROR
        }
    }

############################################################################
```

```
#
#  The Camera Viewing System
#

CameraDistance  = 48.56                  # pre-calculated
MirrorDistance  =  7.00
FocusDistance   =  5.00                  # average

CenterX         =  3.45
CenterY         =  6.33

ZMOV( -( CameraDistance - MirrorDistance ) )
YMOV( -0.77 )                            # mirror off-axis amount
XMOV( 0.25 )

#
#  Flat Folding Mirror
#

{*
YROT( 180 )                              # face forwards
YROT( -45 )                              # point down x-axis
YROT( CenterX / 2 )                      # correct for off-center
XROT( CenterY / 2 )                      # correct for off-center
XROT( 5 )                                # fudge factor
*}
    {
    LABEL( "TV Guider Fold Mirror" )
    STOPRECT( 2 * sqrt(2), 2 )
    FLAT
    MIRROR
    }

#
#  Camera
#

YROT( 90 )
ZMOV( -MirrorDistance )
YROT( 180 )
YROT( 0.35 )                             # fudge factor
XROT( -2.05 )                            # fudge factor
YMOV( 0.12 )
XMOV( 0.06 )
CANON( FocusDistance, CameraDistance )

#
#  The detector
#

PixelSize = 24
ChipWidthX = 1024
ChipWidthY = 1024

ZMOV( -0.004 )                           # focus for multi-slit mask mode
XROT( 0.90 )                             # best tilt for long-slit mode
# XROT( 1.57 )                           # best tilt for direct viewing mode
ZROT( 11.8 )

LABEL( "TV Detector Chip" )
FLAT
STOPRECT( 0.5 * PixelSize * ChipWidthX / 25400,
          0.5 * PixelSize * ChipWidthY / 25400 )
EXIT
```

```
###############################################################################
#
#   Canon 200mm f/1.8 lens description
#
#   Focus is the focal distance to which the Canon lens is set to, as marked
#   on the camera body.  Since the focal surface is not likely to be set at
#   the same distance as for a 35mm camera, this value will not be the focal
#   distance at which the object resides in actual use.
#
#   ObjectDistance is the distance (in inches) to the object from the
#   first camera surface.
#
#   The lens is approximated by aberrationless paraxial surfaces.  I have
#   not put in the correct pupils (which are irrelevant for performance),
#   so the rays inside the lens appear in strange places.
#

DEFINE CANON( Focus, ObjectDistance )
    {
    mmtoinches = 1.0 / 25.4

    Aperture            = 102.5 * mmtoinches
    FlangeDistance      = 195.1 * mmtoinches
    FlangeRadius        =  31.8 * mmtoinches + 0.25     # oversized...

#
#   The following values are measured at 1/Focus = 0.0 meters and
#   1/Focus = 0.4 meters.
#
#

    FocalLength0.0      = 194 * mmtoinches
    FocalLength0.4      = 199 * mmtoinches
    BackPP0.0           = 160 * mmtoinches
    BackPP0.4           = 176 * mmtoinches
    FrontPP0.0          = 200 * mmtoinches
    FrontPP0.4          = 158 * mmtoinches

#
#   The above values are interpolated linearly in 1/Focus.
#

    power = 2.5 / Focus

    FocalLength = (FocalLength0.4 - FocalLength0.0) * power + FocalLength0.0
    BackPP      = (     BackPP0.4 -      BackPP0.0) * power +      BackPP0.0
    FrontPP     = (    FrontPP0.4 -     FrontPP0.0) * power +     FrontPP0.0

    BackFocalDistance = 1 /
        ( 1 / FocalLength - 1 / ( ObjectDistance + FrontPP ) )

PRINT( "BackFocalDistance", BackFocalDistance )

############################################################################
#
#   Now the description...
#

    LABEL( "Canon 200mm f/1.8 Lens -- Aperture and First Surface" )
    FLAT
    STOPCIRCLE( Aperture / 2 )

    ZMOV(0)                                 # surface separator
```

```
LABEL( "Canon 200mm f/1.8 Lens -- Front Principle Plane Distance" )
FLAT
PDIST( FrontPP )

ZMOV(0)                                         # surface separator

LABEL( "Canon 200mm f/1.8 Lens -- Front Thin Lens Approximation" )
PARAX( FocalLength )
FLAT

ZMOV(0)                                         # surface separator

LABEL( "Canon 200mm f/1.8 Lens -- Back Principle Plane Distance" )
FLAT
PDIST( BackPP - FlangeDistance )

ZMOV( FlangeDistance )

LABEL( "Canon 200mm f/1.8 Lens -- Rear Mounting Flange" )
FLAT
STOPCIRCLE( FlangeRadius )

ZMOV( BackFocalDistance - BackPP )

LABEL( "Canon 200mm f/1.8 Lens -- Paraxial Image Position" )
FLAT
}
```

# DEIMOS TV in Offset-Guiding Mode



# DEIMOS TV in Multi-Slit Viewing Mode



DEIMOS TV in Long-Slit Viewing Mode



DEIMOS TV in Offset-Guiding Mode



SUTIN 11/21/96

# DEIMOS TV Vignetting from the Cylindrical Slit Mask



JTW 4/2/96

# DEIMOS TV RMS Spot Diameter from the Cylindrical Slit Mask



11/21/96

DEIMOS TV RMS Spot Diameter from Long Slit Mask



SUTIN 11/21/96

DEIMOS TV RMS Spot Diameter from Offset Guide Mirror



Canon Aperture in Long Slit Mode

Canon Aperture in Offset Guide Mode